
1 はじめに
筆者が子供であった 1970年代はまだ電卓（電子卓上計算機）は珍しい存在であり，計算処理は人間の手で行うもの

であるというのが 1つの常識であった．小学校では四則計算のトレーニングが重視され，珠算の技能を持つものは就
職においても優遇されていた時代である．今では電卓だけでなくパソコン（パーソナルコンピュータ）が十分普及し，
スプレッドシートなどの便利なソフトウェア・ツールも利用でき，計算処理は人間の手作業を離れて，誰もが実行で
きるものになった．計算機が万人のツールとなったことで，手作業による数値計算が苦手な人も計算という技能の恩
恵を浴することができるようになったことは重要な社会的進歩であると考えることができる．
ここで紹介するのは数式処理システムの 1つであるMaximaというソフトウェア・ツールである．数式処理シス

テムも計算処理を実行するシステムであるが，それは数値計算の実行のみならず，記号代数的に表現された数式を記
号のまま計算処理することができるという点が最も重要である．簡単にいうと，

a+ a

という式を
2a

と簡略化する作業を自動的に実行するような処理である．これだけでは数式処理システムの有用性がよくわからない
という場合，(x+ a)30 を展開する例を見てみる．下の例は実際にこの展開作業をMaximaで行ったものである．

x30 + 30 a x29 + 435 a2 x28 + 4060 a3 x27 + 27405 a4 x26 + 142506 a5 x25 + 593775 a6 x24+

2035800 a7 x23 + 5852925 a8 x22 + 14307150 a9 x21 + 30045015 a10 x20 + 54627300 a11 x19+

86493225 a12 x18 + 119759850 a13 x17 + 145422675 a14 x16 + 155117520 a15 x15 + 145422675 a16 x14+

119759850 a17 x13 + 86493225 a18 x12 + 54627300 a19 x11 + 30045015 a20 x10 + 14307150 a21 x9+

5852925 a22 x8 + 2035800 a23 x7 + 593775 a24 x6 + 142506 a25 x5 + 27405 a26 x4 + 4060 a27 x3+

435 a28 x2 + 30 a29 x+ a30

勿論これは筆者が手で計算したものではなく，Maximaが基本的に持っている計算機能により自動的に算出されたも
のであり，計算結果は一瞬にして得られる．

Maximaでは代数式の展開や因子分解（因数分解），微分や積分とそれに関連する解析学的な操作，方程式（代数
方程式，微分方程式）の求解，各種の関数に関する計算，行列とベクトルに関する計算からグラフの描画まで行うこ
とができる．このようなソフトウェア・ツールの普及により高度な数学的計算が手軽に行え，私達の知的活動の範囲
が大幅に広がることが期待できる．

1.1 Maximaの生い立ち

Maximaの開発は 1960年代まで遡ることができる．当初Macsymaという数式処理システムがMIT（マサチュー
セッツ工科大学）でMaclisp（LISP系言語の 1つ）を用いて開発された．それが 1982年に Common Lisp（これも
LISP系言語の 1つ）の言語処理系の上に移植され，Maximaと呼ばれるシステムとなった．現在MaximaはGNUの
GPLに基づくフリーソフトウェアとして配布されている．

1.1.1 wxMaxima

現在は，TEXの数式清書機能を持つフロントエンドを備えた wxMaxima（図 1）というアプリケーションとして利
用でき，

　　 http://maxima.sourceforge.net/

から入手が可能である．

本書では，wxMaxima の操作環境を前提としてMaximaの使用方法について解説する．

1



左がアプリケーションのアイコン，システムを起動すると右のようなウィンドウが表示される．

図 1: wxMaxima

2 基本的な機能

2.1 数式の入力

数式はMaximaのコマンドとして入力する．1つの式の終端には ‘;’（セミコロン）を付け 1 ， Shift + Enter を
押すことで処理が実行される．（ Enter のみの場合は単なる改行処理となる）

算術演算の記号は下記の通りである．

• 和　 ‘+’で記号や式を連結する．
例．a+ aの簡単化

(%i1) a+a;

(%o1) 2 a

• 積　 ‘*’で記号や式を連結する
例．x× xの簡単化

(%i2) x*x;

(%o2) x2

• 差　 ‘-’で記号や式を連結する
例．5a− 2aの簡単化

(%i3) 5*a - 2*a;

(%o3) 3 a

• 商　 ‘/’で記号や式を連結する

例．6y7

4y3
の簡単化

(%i4) 6*y^7 / (4*y^3);

(%o4)
3 y4

2

1最新の wxMaxima では，終端のセミコロンを付けなくても自動的に付加される．

• 冪乗　 ‘ˆ’で記号や式を連結する
上記の例を参照のこと．

2.1.1 ノートブックとセル

wxMaximaのウィンドウ内には入力した式とそれを評価した結果が表示されている．この「入力と評価結果」の組
はセルと呼ばれ，1つの処理の単位となる．複数の処理の過程はノートブックという形式で保持され，セルの連鎖か
ら成る．

2.1.1.1 評価結果の非表示
入力する式の末尾に ‘$’ を付けると，評価結果を表示しない．

2.1.2 処理過程のヒストリ

上の実行例の中に見られる%記号は処理過程のヒストリを意味する．例えば%o2には x2が，%o3には 3aが割り当
てられており，これをそのまま以後の処理に使用することができる．（下記参照）

(%i5) %o2 + %o3;

(%o5) x2 + 3 a

2.2 式の展開と因数分解

式を展開する場合は expand(式)，因数分解する場合は factor(式) を実行する．（下記参照）

(%i6) expand( (a*x^2+b*x+c)*(d*x^2+e*x+f) );

(%o6) a d x4 + a e x3 + b d x3 + a f x2 + b e x2 + c d x2 + b f x+ c e x+ c f

(%i7) factor( % );

(%o7)
(
a x2 + b x+ c

) (
d x2 + e x+ f

)

この例の%i7に factor(%)と入力があるが，1文字の%は直前の処理結果を意味する．

2.3 基本的な関数

Maximaで使える数学的関数の基本的なものを表 1に挙げる．

2.4 重要な定数

表 2に重要な定数を挙げる．これらの値を用いた計算の例を下に示す．

(%i8) sin(%pi/2);

(%o8) 1

(%i9) log(%e);

(%o9) 1

3



表 1: 基本的な関数

記述 意味

sqrt(x)
√
x

log(x) log x 自然対数
exp(x) ex 指数関数
sin(x) sinx 正弦関数
cos(x) cosx 余弦関数
tan(x) tanx 正接関数
asin(x) sin−1 x 正弦関数の逆関数
acos(x) cos−1 x 余弦関数の逆関数
atan(x) tan−1 x 正接関数の逆関数
abs(x) |x| 絶対値

mod(x,p) x mod p 法 pでの xの剰余

表 2: 重要な定数

記述 意味

%pi π（円周率）
%e e（ネイピア数：自然対数の底）
%i i（虚数単位）

(%i10) %i * %i;

(%o10) − 1

上から順に，sin
(π
2

)
, loge e, i2 を計算した例である．

2.5 数値計算

式の値を数値の形で求めるには bfloat(式)を実行する．下に実行例を示す．

(%i11) tan(%pi/3);

(%o11)
√
3

(%i12) bfloat(%);

(%o12) 1.732050807568877b0

(%i13) bfloat(%pi);

(%o13) 3.141592653589793b0

求める数値の精度（桁数）を指定するには，‘fpprec:桁数’を実行する．（下記参照）

(%i14) fpprec:70;

(%o14) 70

(%i15) bfloat(%pi);

(%o15) 3.141592653589793238462643383279502884197169399375105820974944592307816b0

これは計算する精度を 70桁に指定した例である．

2.6 変数への値の設定と関数の定義

コロン ‘:’を使って，変数に値を設定することができる．（下記参照）

(%i17) a:21;

b:57;

(a) 21

(b) 57

(%i18) a+b;

(%o18) 78

これは，変数 aに 21を，変数 bに 57を設定した後，a+ bを計算している例である．

‘:=’を使って，関数を定義することができる．（下記参照）

(%i19) f(x) := 3*x^2+1;

(%o19) f (x) := 3x2 + 1

(%i20) f(5);

(%o20) 76

これは，f(x) = 3x2 + 1を定義した後，f(5)を算出している例である．

変数への値の設定や関数の定義を解除するには kill(変数名もしくは関数名) を実行する．（下記参照）

(%i21) kill(a,b,f);

(%o21) done

(%i23) a+b;

f(5);

(%o22) b+ a

(%o23) f (5)

値の設定と関数の定義が解除されていることがわかる．

kill(all)を実行すると，全ての設定と定義が解除される．

2.7 総和

sum(f(k),k,i,n)を実行することで，式 f(k)の k = i · · ·nの範囲の総和を求めることができる．
1～10の合計を求めるには次のように実行する．

(%i24) sum(k,k,1,10);

(%o24) 55

下のように総和は形式的に処理することもできる．

(%i26) sum(k*d,k,i,n);

sum(r^k,k,i,n);

(%o25) d

n∑
k=i

k

(%o26)
n∑

k=i

rk

2.7.1 総和の一般化

sumの実行時に，simpsum=true 2 を付加することで計算結果を一般化（簡単化）できる場合がある．（下記参照）

(%i28) sum(k*d,k,i,n),simpsum=true;

sum(r^k,k,i,n),simpsum=true;

(%o27) d

(
n2 + n

2
− i+ (i− 1)

2 − 1

2

)

(%o28)
rn+1 − ri

r − 1

2.8 wxMaximaフロントエンド

wxMaximaのフロントエンドウィンドウの基本的な操作方法について説明する．

2.8.1 ノートブックの保存

「ファイル」メニューから「名前を付けて保存」を選択すると，保存作業のためのダイアログが開く．（図 2参照）
ここで保存する場所とファイル名を設定して「保存」ボタンをクリックする．

保存するノートブックのファイル名の拡張子は *.wxmx となる．

参考）計算結果を Common Lispのプログラムとしてファイルに保存するには，save(”ファイル名”,all)を実行する．
（下記参照）

(%i29) save("TheHistory",all);

(%o29) C : /ProgramFiles/Maxima− 5.28.0− 2/wxMaxima/TheHistory

2true は真理値である．偽は false である．

　　→　　

図 2: ノートブックの保存

2.8.2 ノートブックの読込み

保存したノートブックを読み込むには，「ファイル」メニューから「開く」を選択する．この操作によりファイルを
選択するダイアログが開くので，場所とファイルを選択して「開く」ボタンをクリックする．

この操作によってノートブックを読み込んだ段階では，まだその内容は評価されていない．ノートブック内のすべ
てのセルを再度評価（図 3）することで，後続の処理をノートブック内に追加することができる．

「セル」メニューから「すべてのセルを評価」を選択する

図 3: セルの再評価

2.8.3 処理結果の TEXフォーマット出力

ノートブックの内容を TEXの形式で保存することができる．「ファイル」メニューから「エクスポート」を選択する
と，出力作業のためのダイアログが開くので，保存する場所とファイル名，それにファイルの種類として「TeXファ
イル」を設定して「保存」ボタンをクリックする．

3 微分・積分
式（関数）の導関数を求めるには diff(式,変数)を実行する．また，原始関数を求めるには integrate(式,変数)を実

行する．（下記参照）

(%i1) diff( a*x^3+b*x^2+c*x+d, x );

(%o1) 3 a x2 + 2 b x+ c

(%i2) integrate( %, x );

(%o2) a x3 + b x2 + c x

7



これは ax3 + bx2 + cx+ dを変数 xで微分し，それの不定積分を求めた例である．（定数項は省略されている）
下記のように，形式的な関数の式 f(x)の導関数や原始関数を求めることもできる．

(%i3) diff( f(x), x );

(%o3)
d

d x
f (x)

(%i4) integrate( %, x );

(%o4) f (x)

(%i5) integrate( f(x), x );

(%o5)

∫
f (x) dx

(%i6) diff( %, x );

(%o6) f (x)

複雑な関数の導関数や原始関数を求めることもできる．（下記参照）

(%i7) diff( sin(cos(x))*cos(sin(x)), x );

(%o7) − cos (x) sin (cos (x)) sin (sin (x))− sin (x) cos (cos (x)) cos (sin (x))

(%i8) integrate( 1/sin(x), x );

(%o8)
log (cos (x)− 1)

2
− log (cos (x) + 1)

2

3.1 級数展開

式（関数）をテーラー展開するには taylor(式,変数,展開の開始値,展開の次数) を実行する．
下の例は f(x) = e−

x
10 sin(x) + 1と定義された関数（図 4）を f(a)の位置を中心にして展開したものである．

まず，次のようにして関数 f(x) = e−
x
10 sin(x) + 1を定義する．

(%i9) f(x):=%e^(-x/10) * sin(x)+1;

(%o9) f (x) := e
−x
10 sin (x) + 1

関数の定義には ‘:=’を用いる．
5次の項まで計算した例を下に示す．

(%i11) taylor(f(x), x, a, 5);

(%o11)/T/
e

a
10 + sin (a)

e
a
10

− (sin (a)− 10 cos (a)) (x− a)

10 e
a
10

−
(99 sin (a) + 20 cos (a)) (x− a)

2

200 e
a
10

+
(299 sin (a)− 970 cos (a)) (x− a)

3

6000 e
a
10

+

(9401 sin (a) + 3960 cos (a)) (x− a)
4

240000 e
a
10

− (49001 sin (a)− 90050 cos (a)) (x− a)
5

12000000 e
a
10

+ ...

図 4: f(x)

3.2 極限

式（関数）の極限値を求めるには limit(式,変数,求める極限の位置（先）) とする．

上で定義した関数 f(x)の x → ∞の極限を求めた例を下に示す．

(%i12) limit(f(x), x, inf);

(%o12) 1

上記 infは∞を意味する．−∞を求める場合はminf を指定する．
関数の極限は，近づく方向（定義域の左右）によって値が異なることがある．次のような関数を考える．

g(x) =
1

x− 1
+ 1

これをMaximaで次のように定義する．

(%i13) g(x):=1/(x-1)+1;

(%o13) g (x) :=
1

x− 1
+ 1

この関数は図 5のように x = 1で発散するが，この点に近づく方向によって −∞，∞のどちらに発散するかが異
なる．
下に示す例は，極限に近づく方向の左右を指定しているものである．

(%i15) limit(g(x), x, 1, minus);

(%o15) −∞

(%i16) limit(g(x), x, 1, plus);

(%o16) ∞

定義域の左側から極限に近づく場合はminusを，右側から近づく場合は plus を指定する．

図 5: f(x)

4 方程式の求解

4.1 代数方程式の求解

代数方程式の解を求めるには，solve(方程式,求める変数)を実行する．下に，2次の代数方程式
ax2 + bx+ c = 0を xについて解く例を示す．

(%i1) solve(a*x^2+b*x+c=0, x);

(%o1) [x = −
√
b2 − 4 a c+ b

2 a
, x =

√
b2 − 4 a c− b

2 a
]

高次の代数方程式も解くことができる．下の例は ax3 + bx2 + cx+ d = 0を xについて解いたものである．

(%i2) solve(a*x^3+b*x^2+c*x+d=0, x);

(%o2) [x =

(
−
√
3 i

2
− 1

2

)
(√

27 a2 d2 + (4 b3 − 18 a b c) d+ 4 a c3 − b2 c2

2 3
3
2 a2

− 27 a2 d− 9 a b c+ 2 b3

54 a3

) 1
3

−(√
3 i
2 − 1

2

) (
3 a c− b2

)
9 a2

(√
27 a2 d2+(4 b3−18 a b c) d+4 a c3−b2 c2

2 3
3
2 a2

− 27 a2 d−9 a b c+2 b3

54 a3

) 1
3

− b

3 a
,

x =

(√
3 i

2
− 1

2

) (√
27 a2 d2 + (4 b3 − 18 a b c) d+ 4 a c3 − b2 c2

2 3
3
2 a2

− 27 a2 d− 9 a b c+ 2 b3

54 a3

) 1
3

−(
−

√
3 i
2 − 1

2

) (
3 a c− b2

)
9 a2

(√
27 a2 d2+(4 b3−18 a b c) d+4 a c3−b2 c2

2 3
3
2 a2

− 27 a2 d−9 a b c+2 b3

54 a3

) 1
3

− b

3 a
,

x =

(√
27 a2 d2 + (4 b3 − 18 a b c) d+ 4 a c3 − b2 c2

2 3
3
2 a2

− 27 a2 d− 9 a b c+ 2 b3

54 a3

) 1
3

−
3 a c− b2

9 a2
(√

27 a2 d2+(4 b3−18 a b c) d+4 a c3−b2 c2

2 3
3
2 a2

− 27 a2 d−9 a b c+2 b3

54 a3

) 1
3

− b

3 a
]

4.1.1 連立 1次方程式

連立 1次方程式を解くには，linsolve(方程式リスト,変数リスト)を実行する．リストは ‘[ ]’内にコンマ ‘,’で区切っ
て項目を列挙したものである．
2つの未知変数 x, yを含む連立方程式 {

ax+ by = e

cx+ dy = f

を解く例を下に示す．

(%i3) linsolve([a*x+b*y=e, c*x+d*y=f], [x,y]);

(%o3) [x = −d e− b f

b c− a d
, y =

c e− a f

b c− a d
]

4.1.2 高次連立方程式

高次連立方程式を解くには，algsys(方程式リスト,変数リスト)を実行する．
2つの未知変数 x, yを含む高次連立方程式{

x2 + x+ y2 + y = 1

−x2 − 2 ∗ x+ y2 + y = −1

を解く例を下に示す．

(%i4) algsys([x^2+x+y^2+y=1, -x^2-2*x+y^2+y=-1], [x,y]);

(%o4) [[x =
1

2
, y = −

√
2 + 1

2
], [x =

1

2
, y =

√
2− 1

2
],

[x = −2, y = −
√
3 i+ 1

2
], [x = −2, y =

√
3 i− 1

2
]]

4.2 微分方程式の求解

微分方程式を解くには，desolve( 微分方程式, 未知関数 ) 3 を実行する．未知の関数 f(x)を含む微分方程式

d

dx
f(x) = f(x)

を解く例を下に示す．

(%i5) desolve( diff(f(x),x)=f(x), f(x) );

(%o5) f(x) = f(0) ex

微分方程式を解く機能は desolve以外にも ode2(微分方程式,未知関数,変数) 4 などがある．

3desolve はラプラス変換を用いて微分方程式を解く．
42 階までの常微分方程式を解く．

4.3 差分方程式の求解（漸化式の一般化）

Maximaには差分方程式（階差方程式）の解を求める（漸化式の一般化を行う）ためのパッケージ ‘solve rec’が提
供されている．この機能を利用するには，下記のような操作により，必要なパッケージをMaximaに読み込んでおく
必要がある．

(%i6) load(solve_rec)$

差分方程式（階差方程式）の解を求めるには，
　　　 solve rec( 差分方程式, 一般化したい項, 初期値, · · · )
を実行する．下に差分方程式 y(n) = (1 + d)y(n− 1)を初期値 y(0) = y0として求める例を示す．

(%i7) solve_rec( y(n)=(1+d)*y(n-1), y(n), y(0)=y0 );

(%o7) y (n) = (d+ 1)
n
y0

方程式は，下記のように添字付きの項（配列）で表現されるものとして記述することもできる．

(%i8) solve_rec( y[n]=(1+d)*y[n-1], y[n], y[0]=y0 );

(%o8) yn = (d+ 1)
n
y0

参考）　差分方程式の解の極限を求めることで，微分方程式の解と同じものを算出することができる場合がある．
　　　（下記参照）

(%i9) limit( %o8, n, inf ),d=x/n;

(%o9) lim
n→∞ yn = ex y0

12



5 ベクトル・行列

5.1 基本

下記のような行列とベクトルに対する処理を例に挙げて説明する．

m1 =

⎛
⎜⎜⎝

1 3 0

0 10 0

13 0 17

⎞
⎟⎟⎠ , m2 =

⎛
⎜⎜⎝
2 0 6

0 9 11

0 0 18

⎞
⎟⎟⎠ , v =

(
a b c

)
, vt =

⎛
⎜⎜⎝
x

y

z

⎞
⎟⎟⎠

行列を構成するには，matrix(行成分リスト 1,行成分リスト 2,· · ·) を実行する．またベクトルは成分のリストで表現
する．（ここでは全て行列として扱う．）

(% i4) m1: matrix(

[1,3,0],

[0,10,0],

[13,0,17]

);

m2: matrix(

[2,0,6],

[0,9,11],

[0,0,18]

);

v:matrix([a,b,c]);

vt:transpose(matrix([x,y,z]));

(m1)

⎛
⎜⎜⎝

1 3 0

0 10 0

13 0 17

⎞
⎟⎟⎠

(m2)

⎛
⎜⎜⎝
2 0 6

0 9 11

0 0 18

⎞
⎟⎟⎠

(v)
(
a b c

)

(vt)

⎛
⎜⎜⎝
x

y

z

⎞
⎟⎟⎠

これで行列とベクトルの定義ができた．上の例の中の transpose(行列) は与えた行列の転置行列を求めるもので，横
並びの成分で構成されるベクトルを縦の形式にする場合にも使用する．

例．行列m1の転置行列を求めるには次のように実行する．

(%i5) transpose(m1);

(%o5)

⎛
⎜⎜⎝
1 0 13

3 10 0

0 0 17

⎞
⎟⎟⎠

5.2 行列の積

行列の積はドット ‘.’で表現する．（下記の例を参照）

(%i6) m1.m2;

(%o6)

⎛
⎜⎜⎝

2 27 39

0 90 110

26 0 384

⎞
⎟⎟⎠

(%i7) m1.vt;

(%o7)

⎛
⎜⎜⎝

3 y + x

10 y

17 z + 13x

⎞
⎟⎟⎠

(%i8) v.vt;

(%o8) c z + b y + a x

5.3 行列式

行列式を求めるには determinant(行列)を実行する．行列m1の行列式を求める例を下に示す．

(%i9) determinant(m1);

(%o9) 170

5.4 逆行列

逆行列を求めるには invert(行列)を実行する．行列m1の逆行列を求める例を下に示す．

(%i10) invert(m1);

(%o10)

⎛
⎜⎜⎝

1 − 3
10 0

0 1
10 0

− 13
17

39
170

1
17

⎞
⎟⎟⎠

5.5 固有値，固有ベクトル

行列の固有値を求めるには eigenvalues(行列) を実行する．固有値だけでなく，固有ベクトルまで求める場合は
eigenvectors(行列) を実行する．
行列m1の固有値と固有ベクトルを求める例を下に示す．

(%i11) ev:eigenvalues(m1);

(ev) [ [1, 10, 17], [1, 1, 1] ]

処理結果は [固有値リスト,重複度リスト]の形で得られる．

(%i12) ev2:eigenvectors(m1);

(ev2) [ [ [1, 10, 17], [1, 1, 1] ], [ [ [1, 0,−13

16
] ], [ [1, 3,−13

7
] ], [ [0, 0, 1] ] ] ]

処理結果は [[固有値リスト,重複度リスト],固有ベクトルのリストのリスト]の形で得られる．
「固有ベクトルのリストのリスト」は各固有値に対応する固有ベクトルの集合をリストにしたものである．

5.6 その他

余因子行列を求めるには，adjoint(行列)を実行する．行列m1の余因子行列を求める例を下に示す．

(%i13) adjoint(m1);

(%o13)

⎛
⎜⎜⎝

170 −51 0

0 17 0

−130 39 10

⎞
⎟⎟⎠

6 グラフィックス

6.1 2次元プロット

y = f(x)（xは独立変数，yは従属変数）という関数がある場合，それをグラフとして可視化することができる．

plot2d(関数の式,[独立変数,開始値,終了値])を実行することでグラフが描画される．下の例は sin(x)を−4 ≤ x ≤ 4

の範囲で描画（図 6）するものである．（wxplot2dを使うとノートブックにグラフが埋め込まれる）

(%i1) plot2d(sin(x),[x,-4,4]);

(%o1)

-1
-0.8

-0.6

-0.4

-0.2

 0

 0.2

 0.4

 0.6

 0.8

 1

-4 -3 -2 -1  0  1  2  3  4

s
i
n

(x
)

x

図 6: 描画結果

下の例のようにして実行することで，従属変数の範囲を指定して描画（図 7）することもできる．

(%i2) plot2d(sin(x),[x,-20,20],[y,-2,2]);

(%o2)

-2

-1.5

-1

-0.5

 0

 0.5

 1

 1.5

 2

-20 -15 -10 -5  0  5  10  15  20

s
i
n

(x
)

x

図 7: 描画結果（縦軸の範囲指定）

同一の座標平面に複数の関数を同時に描画する場合は，下の例のように関数のリストを与えて実行する．

(%i3) plot2d([sin(x),cos(x),exp(x/10)],[x,-10,10],[y,-2,3]);

(%o3)

これは sin(x), cos(x), exp
(

x
10

)
を同時に描画するもので，図 8のようなグラフができる．

-2

-1

 0

 1

 2

 3

-10 -5  0  5  10

x

sin(x)

cos(x)

%e^(x/10)

図 8: 複数の関数の同時描画

6.1.1 座標リストのプロット

下記のように discrete を指定することで，座標リストのプロットが可能となる．
　　　　 plot2d([discrete,リスト], オプション)

　　　 wxplot2d([discrete,リスト], オプション)

(%i7) p:[[0,0],[1,1],[2,2],[3,3],[4,4]]$

plot2d([discrete,p],[style,points],[point_type,circle]);

(%o8)

この例のように，スタイルとして [style,points] を追加することで，点表示のプロットができる．またことのき，
point type オプションを追加することで，点表示のスタイルを変更できる．（図 9の例を参照）

図 9: 点表示のプロット

point typeオプションに与えるスタイルの一部を表 3に挙げる．

表 3: point typeオプションの代表的なもの

スタイル 形状

bullet ●
circle ○
plus ＋
times ×
asterisk ＊
box ■

square □

6.2 3次元プロット

z = f(x, y)（x, y は独立変数，z は従属変数）という関数がある場合，それを 3次元のグラフとして可視化するこ
とができる．

plot3d(関数の式,[独立変数 1,開始値,終了値],[独立変数 2,開始値,終了値]) を実行することで 3次元のグラフが描画
される．（wxplot3d を使うとノートブックにグラフが埋め込まれる）

下の例は cos
(√

x2 + y2
)
を −4 ≤ x ≤ 4,−4 ≤ y ≤ 4の範囲で描画するものである．

(%i4) plot3d(cos(sqrt(x^2+y^2)),[x,-4,4],[y,-4,4]);

(%o4)

この結果，図 10のようなグラフができる．

-4 -3 -2 -1  0  1  2  3  4-4
-3

-2
-1

 0
 1

 2
 3

 4

-1
-0.8
-0.6
-0.4
-0.2

 0
 0.2
 0.4
 0.6
 0.8

 1

cos(sqrt(y^2+x^2))

x

y

z

図 10: 3次元のグラフの描画

下の例のようにして実行することで，描画のメッシュを増やして描画（図 11）することもできる．

(%i5) plot3d(cos(sqrt(x^2+y^2)),[x,-4,4],[y,-4,4],[grid,100,100]);

(%o5)

この例のように，[grid,独立変数 1のメッシュの数,独立変数 2のメッシュの数]を実行時に指定する．

図 11: 3次元のグラフの描画（メッシュ：100× 100）

6.3 wxMaximaの場合の描画

wxMaximaの場合，plot2d,plot3dの代わりにそれぞれwxplot2d,wxplot3dを実行することもできる．この場合，描
画結果のグラフはノートブック内に表示される．

6.3.1 描画時のオプション：　色指定　（wxMaxima＋ gnuplotエンジン）

色を指定してグラフを描画するには color オプションを追加する．
例えば “[color,色]”を追加して，グラフ描画の色を変えることができる．

例．plot2d(sin(x),[x,-4,4],[color,black])

これで描画色が黒になる．MS-Windows環境下での色指定を表 4に挙げる．

表 4: 色指定（MS-Windowsの場合）

black 黒
blue 青
red 赤
green 緑

magenta マゼンタ
cyan シアン

6.3.2 描画時のオプション：　縦横比の設定　（wxMaxima＋ gnuplotエンジン）

オプションとして
　　　　 [gnuplot preamble,”set size 横の比率,縦の比率”]

を追加することで，描画するグラフ領域の縦横の比率を変更することができる．

6.4 画像のファイル出力（wxMaxima＋ gnuplotエンジン）

wxMaximaで plot2dや plot3dを実行すると，描画エンジンが gnuplotの場合，描画結果のグラフが別のウィンド
ウで表示される．この場合，グラフが表示されているウィンドウに「Options」ボタン（メニュー）があり，メニュー
項目「Save as EMF」を選択することでグラフを EMF形式の画像ファイルとして保存することができる．

7 プログラミング
Maximaでは関数定義を行うことでユーザが独自に機能を拡張することができる．ここでは更に，条件判定や繰り

返しなどの制御を実現する方法について説明する．

7.1 条件判定

文法：　 if　条件　 then　式 1　 else　式 2

与えた条件が真である場合に式 1を実行し，条件が偽の場合は式 2を実行する．
下の例は，条件判定を用いてフィボナッチ数列

f(n) =

⎧⎪⎪⎨
⎪⎪⎩

n = 0 → 1

n = 1 → 1

それ以外 → f(n− 1) + f(n− 2)

を計算する関数を定義して実行している例である．

(%i1) f(n) := if n=0 then 1 else if n=1 then 1 else f(n-1)+f(n-2);

(%o1) f (n) := if n = 0 then 1 else if n = 1 then 1 else f (n− 1) + f (n− 2)

(%i2) f(22);

(%o2) 28657

7.1.1 条件式（述語）

条件として記述できるもの（一部）を表 5に挙げる．

表 5: 基本的な述語

記述 意味
X = Y X と Y が等しい
X#Y X と Y は等しくない
X > Y X は Y より大きい
X >= Y X は Y 以上
X < Y X は Y より小さい
X <= Y X は Y 以下
evenp(X) X は偶数
oddp(X) X は奇数
P and Q P かつ Q

P or Q P または Q

not(P ) P でない

7.2 手続き

文法：　 block( [局所変数リスト], 式 1, 式 2, · · · )

式 1, 式 2, · · ·
と順番に実行してゆく方法である．局所変数は当該 blockを実行する範囲内でのみ有効なものである．block文実行後
は最後の式の値が戻り値となる．

(%i3) block([c:0],print(c),c:c+1,print(c),c:c+1,print(c));

0
1
2

(%o3) 2

これは cの値を変えながら print関数で表示している例である．

7.3 繰り返し

文法 1：　 for 制御変数:初期値 step 増分 thru 終値 do 式
文法 2：　 for 制御変数:初期値 step 増分 while 条件式 do 式
文法 3：　 for 制御変数 in [リスト] do 式

文法 1の例．cの値を 1から 5まで 1づつ増やしながらそれを印字する．

(%i4) for c:0 step 1 thru 5 do print(c);

0
1
2
3
4
5

(%o4) done

文法 2の例．cの値を 1づつ増やしながら cが 5未満である間，それを印字する．

(%i5) for c:0 step 1 while c < 5 do print(c);

0
1
2
3
4

(%o5) done

文法 3の例．cにリスト [v,w,x,y,z]の要素を順番に与えながら印字する．

(%i6) for c in [v,w,x,y,z] do print(c);

v
w
x
y
z

(%o6) done

7.4 リスト処理

プログラミングのためのデータ構造としてリストが有用である．

7.4.1 リストの要素の取り出し

リスト Lの n番目の要素を取り出すには，L[n] を評価する．

(%i7) for c:1 step 1 thru 4 do print([s,t,u,v][c]);

s
t
u
v

(%o7) done

これは，リスト [s,t,u,v]の要素を順番に表示する例である．

7.4.2 リストの合成と分解

● cons　リスト Lの先頭に要素 aを追加したものを得るには，cons(a,L) を評価する．

(%i8) cons(a,[b,c,d]);

(%o8) [a, b, c, d]

これは，リスト [b,c,d]の先頭に要素 aを追加したものを得る例である．

● endcons　要素 dをリスト Lの末尾に追加するには，endcons(d,L) を評価する．

(%i9) endcons(d,[a,b,c]);

(%o9) [a, b, c, d]

これは，リスト [a,b,c]の末尾に要素 dを追加したものを得る例である．

● first　リスト Lの先頭の要素を取り出すには，first(L) を評価する．

(%i10) first([a,b,c,d]);

(%o10) a

これは，リスト [a,b,c,d]の先頭の要素を得る例である．

● rest　リスト Lの先頭の要素を取り除いたリストを得るには，rest(L) を評価する．

(%i11) rest([a,b,c,d]);

(%o11) [b, c, d]

これは，リスト [a,b,c,d]の先頭の要素を取り除いたリスト得る例である．

● last　リスト Lの末尾の要素を得るには，last(L) を評価する．

(%i12) last([a,b,c,d]);

(%o12) d

これは，リスト [a,b,c,d]の末尾の要素を得る例である．

● append　リスト L1と L2を連結するには，append(L1,L2) を評価する．

(%i13) append([a,b],[c,d]);

(%o13) [a, b, c, d]

これは，リスト [a,b]と [c,d]を連結したリストを得る例である．

7.4.3 リストの長さの算出

リスト Lの長さを算出するには，length(L) を評価する．

(%i14) length([a,b,c,d,e]);

(%o14) 5

これは，リスト [a,b,c,d,e]の長さを算出した結果，5が得られている例である．

7.4.4 空リストの判定

リスト Lが空であるかどうかを判定するには，emptyp(L) を評価する．Lが空の場合は trueを，空でない場合は
falseを返す．（下記参照）

(%i15) emptyp([]);

emptyp([a]);

(%o15) true

(%o16) false

7.5 プログラムの作成と読込み（wxMaxima）

関数の定義や値の設定などをテキストファイルの形で保存しておき，それをMaximaに読み込んで使用することが
できる．

下のようなプログラム（テキスト形式）が fb.macというファイル名で保存されているとする．wxMaximaのフロン
トエンドウィンドウのメニュー「ファイル」から「パッケージ読込み」を選択すると，ファイルを選択するダイアログ
が表示される．ここでプログラムのファイルを選択して「開く」ボタンをクリックするとプログラムが読み込まれる．

プログラム：fb.mac

1 /* フ ィ ボ ナ ッ チ 数 */

2 f(n) :=

3 if n=0 then 1 else

4 if n=1 then 1 else

5 f(n-1)+f(n-2);

このプログラム 5 を読み込んだ後，関数 fが使用できる．実行例を下に示す．

(%i1) load("D:/fb.mac")$

(%i2) f(0);f(1);f(2);f(3);f(4);

(%o2) 1

(%o3) 1

(%o4) 2

(%o5) 3

(%o6) 5

この実行例の冒頭で load関数が呼び出されているが，「ファイル」メニューから「パッケージ読込み」を選択する操作
によって自動的に実行されるものである．

5このプログラムのアルゴリズムは最適ではない．計算に必要となる記憶域と時間に関して更に効率の良いアルゴリズムが存在する．

8 その他

8.1 Maxima⇔Lisp処理系の切り替え

Maximaは Lisp処理系（Common Lisp）の上に構築されており，Lisp処理系の機能を使うことができる．具体的
にはMaximaの式 to lisp() を評価すると Lisp処理系に制御が移る．また Lisp処理系からMaximaに戻るには Lisp

の S式 (to-maxima) を評価する．

(%i1) to lisp();

Type (to-maxima) to restart, ($ quit) to quit Maxima.

MAXIMA> (car ’(a b c d e));
A

MAXIMA> (to-maxima)

Returning to Maxima

(%o1) true

(%i2) a+a;

(%o2) 2a

上の例の (%i1) で to lisp() を評価して Common Lisp処理系に切り替えている．この結果，プロンプトが

　MAXIMA＞

に変わり，Common Lisp の式の入力を受け付ける．この状態で S式 (to-maxima) を評価するとMaximaに制御が戻
り（上記例の%i2～%o2），以後はMaximaの式を受け付ける．

